2．開水路における等流の計算法（ここでは，広矩形断面とみなし径深を h とする）

2－1　矩形断面

下の図２－１－１矩形断面について

$n=0.02$，$I=1/100$

の条件で流速，流量を求める



図２－１－１　矩形断面

①流速はこの場合流量がわからないのでマニングの式式（１－３－４）を用いる。

式（１－３－４）

$$V=\frac{1}{n}h^{2/3}I^{1/2}$$

$$=\frac{1}{0.02}\times 1^{2/3}\times\left(\frac{1}{100}\right)^{1/2}$$

$$=5(\mathrm{m/s})$$

②流量は式（１－３－２）を使う。

$$Q=AV$$

$$=1\times 20\times 5$$

$$=100(\mathrm{m^3/s})$$

したがって，図２－１－１の流量は $100\mathrm{m^3/s}$ となる。

2 － 2　任意形状断面

2 － 2 － 1　任意形状断面 1 （流量を求める）

図 2 － 2 － 1 の断面について水深 1 m 側が　$n_1=0.04$，3 m 側が　$n_2=0.02$，$I=1/100$ の条件で流量を求めよ。



図 2 － 2 － 1　任意形状断面

①この断面は矩形断面ではないが，図 2 － 2 － 2 の様に［1］，［2］断面にそれぞれ分けておのおのの　A_1V_1，A_2V_2 を求める。



図 2 － 2 － 2

$A_1=1\times10$　　　　$A_2=3\times10$

$=10(\mathrm{m}^2)$　　　　$=30(\mathrm{m}^2)$

式（1 － 3 － 4）より

$$V_1=\frac{1}{n_1}h_1^{2/3}I^{1/2} \qquad V_2=\frac{1}{n_2}h_2^{2/3}I^{1/2}$$

$$=\frac{1}{0.04}\cdot1^{2/3}\cdot\left(\frac{1}{100}\right)^{1/2} \qquad =\frac{1}{0.02}\cdot3^{2/3}\cdot\left(\frac{1}{100}\right)^{1/2}$$

$$=2.5(\mathrm{m/s}) \qquad =10.4(\mathrm{m/s})$$

②［1］，［2］断面でそれぞれ求めと A_1，V_1，A_2，V_2 で断面全体の流量を求める。

式（1 － 3 － 2）より

$$Q=A_1V_1+A_2V_2$$
$$=10\cdot2.5+30\cdot10.4$$
$$=25+312$$
$$=337(\mathrm{m}^3/\mathrm{s})$$

2－2－2　任意形状断面2（水深を求める）



図2－2－2　任意断面

図2－2－3の断面について下記の条件で水深 h_1 を求める。

h_1 側 $n_1=0.02$, h_2 側 $n_2=0.04$

$I=1/1000$,　$Q=80m^3/s$,　$h_1=h_2-1$

①2－2－1と同じように断面を分けて考える。図2－2－2参照

$$h_2=h_1+1$$

$$A_1=10h_1 \qquad A_2=10h_2=10(h_1+1)$$

式（1－3－4）より

$$V_1=\frac{1}{n_1}h_1^{2/3}I^{1/2} \qquad V_2=\frac{1}{n_2}h_2^{2/3}I^{1/2}$$

$$\frac{1}{0.02}h_1^{2/3}\left(\frac{1}{1000}\right)^{1/2} \qquad =\frac{1}{0.04}(h_1+1)^{2/3}\left(\frac{1}{1000}\right)^{1/2}$$

$$=1.58h_1^{2/3} \qquad =0.79(h_1+1)^{2/3}$$

$$Q=A_1V_1+A_2V_2$$
$$=10h_1\cdot1.58h_1^{2/3}+10(h_1+1)\cdot0.79(h_1+1)^{2/3}$$
$$=15.8h_1^{5/3}+7.9(h_1+1)^{5/3}$$
$$=80$$

②前式 Q を近似解法により h_1 を求める。近似解法については、ニュートン法を用いる。(ニュートン法については補遺参照)

Q の式を $f_{(h1)}$ の式にする。

$$f_{(h1)}=15.8h_1^{5/3}+7.9(h_1+1)^{5/3}-80$$

$f_{(h1)}$ を1回微分する。

$$f'_{(h1)}=26.3h_1^{2/3}+13.2(h_1+1)^{2/3}$$

修正値 Δh を出す。

$$\varDelta h = -\frac{f_{(h1)}}{f'_{(h1)}}$$

$$= -\frac{15.8h_1^{5/3} + 7.9(h_1+1)^{5/3} - 80}{26.3h_1^{2/3} + 13.2(h_1+1)^{2/3}}$$

h_1の初期値を１ｍとして近似を進めていくと表２－２－１のようになる。

表２－２－１　近似計算表

h_1	$f_{(h1)}$	$f'_{(h1)}$	$\varDelta h$
1 m	-39.12	47.25	0.83
1.83	7.99	65.76	-0.12
1.71	0.25	63.27	0.004
1.706			

2－3　流量・水位曲線の作成方法

台形断面を使った $h-Q$ 曲線をつくる。台形断面については，図２－３－１を考える。



h：水深

B：河川底の幅

B_u：河川水面での幅

γ：河側勾配割合

L：河川側面長

図２－３－１　台形断面

$h-Q$ 曲線とは流量 Q が一定の変化量で増加する時に，水深 h がどの様に変化するかを表したものである。

普通，流量が増えれば水深が高くなるのは当たり前であるが，どの様に変化していくのか分かるだろうか？その事を明らかにするのが，この問題の目的の目的である。まず，Q に任意の値を与えられた時の h の算出法を考えてみる。

例題を２つ挙げる。

例題　２－３－１　はさみうち法を使った方法（はさみうち法については，補遺参照）

$$L=h\sqrt{1+\gamma^2} \cdots\cdots ①$$

$$S=B+2L \cdots\cdots ②$$

$$Bu=B+2h\gamma \cdots\cdots ③$$

$$A=\frac{1}{2}(B+Bu)h \cdots\cdots ④$$

式（1－3－1）より

$$R=\frac{A}{S} \cdots\cdots ⑤$$

式（1－3－3）より

$$V=\frac{1}{n}R^{2/3}I^{1/2} \cdots\cdots ⑥$$

式（1－3－2），⑥より

$$Q=AV$$

$$=A\cdot\frac{1}{n}R^{2/3}I^{1/2} \cdots\cdots ⑦$$

$A\cdot\frac{1}{n}R^{2/3}I^{1/2}$ を Q_1 とする。

h を仮定すると、L， S，Bu， A， R， V の値がでてくるので，Q_1 を計算する。

もし，$Q_1=Q$ ならば，その Q_1 の h が答え。

もし，Q_1 が Q より小さいならば，h を大きくして，再び計算を繰り返す。

もし，Q_1 が Q より大きいならば，h を小さくして，再び計算を繰り返す。

図２－３－２　$h-Q$ プログラム１のフロチャート



表２－３－１　*h*－*Q* プログラム１

```
1000  DIM B(250),Z(250),N(250),H(250),M(250)
1010 Q=20:B=10:I=.001:N=.02:ALF=1:M=0
1020 HMAX=100
1030 HMIN=0
1040 M=0
1050 LPRINT "H=";H
1060 H=(HMIN+HMAX)/2
1070 M=M+1
1080 EPS=.01
1090 L=H*SQR(1+ALF^2)
1100 S=B+2*L
1110 BU=B+2*H*ALF
1120 A=.5*(B+BU)*H
1130 R=A/S
1140 Q1=R^(2/3)*I^(1/2)*A/N
1150 IF ABS(Q-Q1)<EPS GOTO 1180
1160 IF Q1<Q THEN HMIN=H : GOTO 1050
1170 IF Q1>Q THEN HMAX=H : GOTO 1050
1180 LPRINT "H=";H,"Q="Q,"Q1=";Q1,"M=";M
```

以上の結果表２－３－２のような値が出る。

表２－３－２　*h*－*Q* プログラム１の結果

```
H= 0
H= 50
H= 25
H= 12.5
H= 6.25
H= 3.125
H= 1.5625
H= .78125
H= 1.17188
H= .976563
H= 1.07422
H= 1.12305
H= 1.14746
H= 1.15967
H= 1.15356
H= 1.15662
H= 1.15509
H= 1.15585     Q= 20          Q1= 20.0024    M= 17
```

表2－3－3　$h-Q$ プログラム1の説明

1000：B，Z，N，h のディメンジョン
1010：初期値
1020：$hmax$ の初期設定
1030：$hmin$ の初期設定
1040：計算回数のカウントの初期値
1060：$hmax$ と $hmin$ の中間値の計算
1070：計算回数のカウント
1080：収束値
1090
～：L，S，Bu，A，R の計算
1130
1140：Q_1 の計算
1150
～：Q_1 の判定
1170
1180：計算結果の出力

$Q=20\,m^3/s$ での水深 h は，1.1585 m で，計算回数 M は17回というのが最終結果となる。実際流量 Q と計算流量 Q_1 との最終的な誤差が0.024となり，収束値 ε の条件を満たしている。

例題　２－３－２　ニュートン法を使った方法（ニュートン法については，補遺参照）台形断面は２－３－１と同じ

$$L=h\sqrt{1+\gamma^2} \cdots\cdots ①$$

①より

$$S=B+2L \cdots\cdots ②$$

$$=B+2h\sqrt{1+\gamma^2}$$

$$Bu=B+2\gamma h \cdots\cdots ③$$

$$A=Bh+\gamma h^2 \cdots\cdots ④$$

式（１－３－１），②，④より

$$R=\frac{\mathrm{A}}{\mathrm{S}}=\frac{Bh+\gamma h^2}{B+2h\sqrt{1+\gamma^2}} \cdots\cdots ⑤$$

式（１－３－３），⑤より

$$V=\frac{1}{n}R^{2/3}\ I^{1/2}$$

$$=\frac{1}{n}\left(\frac{Bh+\gamma h^2}{B+2h\sqrt{1+\gamma^2}}\right)^{2/3} I^{1/2} \cdots\cdots ⑥$$

式（１－３－２），④，⑥より

$$Q=AV$$

$$=(Bh+\gamma h^2)\left\{\frac{1}{n}\left(\frac{Bh+\gamma h^2}{B+2h\sqrt{1+\gamma^2}}\right)^{2/3} I^{1/2}\right\} \cdots\cdots ⑦$$

$$f_{(h)}\ =Q-(Bh+\gamma h)^2\left\{\frac{1}{n}\left(\frac{Bh+\gamma h^2}{B+2h\sqrt{1+\gamma^2}}\right)^{2/3} I^{1/2}\right\}$$

$$=Q-\frac{(Bh+\gamma h^2)^{5/3}}{(B+2h\sqrt{1+\gamma^2})^{2/3}}\ \frac{I^{1/2}}{n} \cdots\cdots ⑧$$

$f_{(h)}=0$ を満たす h が答え。

$f'_{(h)}$ が求まれば，

$$\Delta h=-\frac{f_{(h)}}{f'_{(h)}}$$

が出来る。

⑧を次のように変える。

$$f_{(h)} = Q - \frac{P_{(h)}}{G_{(h)}}\left(\frac{I^{1/2}}{n}\right) \cdots\cdots ⑨$$

$$P_{(h)} = (Bh + \gamma h^2)^{5/3}$$

$$G_{(h)} = (B + 2h\sqrt{1+\gamma^2})^{2/3}$$

$$f'_{(h)} = \frac{G_{(h)} \times P'_{(h)} - G'_{(h)} \times P_{(h)}}{G^2_{(h)}}\left(\frac{I^{1/2}}{n}\right) \cdots\cdots ⑩$$

$$P'_{(h)} = \frac{5}{3}(Bh + \gamma h^2)^{2/3}(B + 2\gamma h)$$

$$G'_{(h)} = \frac{2}{3}(B + 2h\sqrt{1+\gamma^2})^{-1/3}(2\sqrt{1+\gamma^2})$$

$f'_{(h)}$の証明

$$f_{(x)} = \frac{f_{2(x)}}{f_{1(x)}}$$

$$f_{1(x)}\, f_{(x)} = f_{2(x)}$$

両辺を x で微分

$$f_{1(x)}\, f'_{(x)} + f'_{1(x)}\, f_{(x)} = f'_{2(x)}$$

$$\therefore f'_{(x)} = \frac{f'_{2(x)} - f'_{1(x)}\, f_{(x)}}{f_{1(x)}} = \frac{f'_{2(x)} - f'_{1(x)}\, \frac{f_{2(x)}}{f_{1(x)}}}{f_{1(x)}} = \frac{f_{1(x)}\, f'_{2(x)} - f'_{1(x)}\, f_{2(x)}}{{f_{1(x)}}^2}$$

これをコンピュータで答を求める。



図2－3－3　h－Q プログラム2のフロチャート

表２－３－３　h－Q プログラム２

```
1000 NI=100:DIM H(NI),P(NI),G(NI)
1010 Q=20:I=.001:N=.02:B=10:ALF=1:X=0
1020 H=1
1030 EPS=.01
1040 X=X+1
1050 LPRINT "H=";H
1060 P=(B*H+ALF*H^2)^(5/3)
1070 G=(B+2*H*SQR(1+ALF^2))^(2/3)
1080 DP=5/3*(B*H+ALF*H^2)^(2/3)*(B+2*ALF*H)
1090 DG=2/3*(B+2*H*SQR(1+ALF^2))^(-1/3)*(2*SQR(1+ALF^2))
1100 FH=Q-P/G*I^(1/2)/N
1110 DFH=-I^(1/2)/N*(G*DP-DG*P)/G^2
1120 DH=-FH/DFH
1130 QC=Q-FH
1140 IF ABS(FH)<EPS GOTO 1170
1150 H=H+DH
1160 GOTO 1040
1170 LPRINT "X=";X,"H=";H,"QC=";QC
```

以上の結果　表２－３－４のような値が出る。

表２－３－４　h－Q プログラム２の結果

```
H= 1
H= 1.16399
H= 1.15579
X= 3          H= 1.15579     QC= 20.0006
```

表２－３－５　$h-Q$ プログラムの説明

1000：h，P，Gのディメンジョン
1010：初期値の設定
1020：hの初期値
1030：収束値
1040：計算回数のカウント
1060
～：計算式
1130
1140：$f_{(h)}$の判定式
1150：hの修正
1170：計算結果の出力

$Q=20\,m^3/s$ での水深 h は1.15579 m で、計算回数 X は３回というのが最終結果となる。実際流量 Q と，計算流量 QC との最終的な誤差が0.0006となり，収束値ε の条件を満たしている。

(1)と(2)の結果を見てもらえば分かると思うが，２つの方法に計算回数の違いこそあるが，結果については，ほとんど無視できるほどの誤差しかない。以上の結果から，どちらの方法を使っても，でてくる数値は同じだといえる。どちらの方法を使うかは，自らの好みではあるが，まだほかに別の方法があると思うので，そちらの方も考えてみたらいいだろう。

2－4　流量・水深曲線

(1)(2)で任意流量での、水深のだし方が分かったと思うので、次に初めの目的であった、流量を変化させたときの水深の変化について計算を行う。

ここでは、前述のニュートン法でのプログラムを使用する。

考え方は、プログラムを流量の変化毎に計算させて、その結果を出力させる。

最後に、出力結果をデータとして、$h-Q$曲線図を作る。このとき、Qを 2 m^3/s から100 m^3/s まで、 2 m^3/s ずつ増加させていくものとする。



図２－４－１　$h-Q$ 曲線フローチャート

表2－4－1　$h-Q$ 曲線プログラム

```
10 'save "B:DAIKEI3.BAS",A
100 DIM H(100),Q(100)
200  OPEN"B:HQ.DAT"FOR OUTPUT AS #1
300 FOR I=1 TO 50
400 Q(I)=I*2
1010 IN=.001:N=.02:B=10:ALF=1:X=0
1020 H(I)=1
1030 EPS=.01
1035 'X=X+1
1036 'PRINT "Q(I)=";Q(I),"H(I)=";H(I)
1040 P=(B*H(I)+ALF*H(I)^2)^(5/3)
1050 G=(B+2*H(I)*SQR(1+ALF^2))^(2/3)
1060 DP=5/3*(B*H(I)+ALF*H(I)^2)^(2/3)*(B+2*ALF*H(I))
1070 DG=2/3*(B+2*H(I)*SQR(1+ALF^2))^(-1/3)*(2*SQR(1+ALF^2))
1080 FH=Q(I)-P/G*IN^(1/2)/N
1090 DFH=-IN^(1/2)/N*(G*DP-DG*P)/G^2
1100 DH=-FH/DFH
1110 QC=Q(I)-FH
1120 IF ABS(FH)<EPS GOTO 1140
1125 H(I)=H(I)+DH
1130 GOTO 1035
1140  PRINT #1,USING"###.###";Q(I),H(I)
1150 'LPRINT USING"###.###";Q(I),H(I)
1500 NEXT I
1510  CLOSE #1
```

以上の事より次のような計算結果ができる。この結果を元に $h-Q$ 曲線図を作る。

表２－４－３　h－Q 曲線プログラムの結果

2.000	0.291	52.000	2.036
4.000	0.440	54.000	2.081
6.000	0.562	56.000	2.126
8.000	0.668	58.000	2.170
10.000	0.763	60.000	2.214
12.000	0.851	62.000	2.257
14.000	0.934	64.000	2.299
16.000	1.012	66.000	2.340
18.000	1.085	68.000	2.381
20.000	1.156	70.000	2.422
22.000	1.223	72.000	2.462
24.000	1.289	74.000	2.501
26.000	1.351	76.000	2.540
28.000	1.412	78.000	2.579
30.000	1.471	80.000	2.617
32.000	1.529	82.000	2.654
34.000	1.585	84.000	2.692
36.000	1.639	86.000	2.728
38.000	1.693	88.000	2.765
40.000	1.745	90.000	2.801
42.000	1.796	92.000	2.836
44.000	1.845	94.000	2.872
46.000	1.894	96.000	2.907
48.000	1.942	98.000	2.941
50.000	1.990	100.000	2.976

表２－４－２　$h-Q$ 曲線プログラムの説明

1000：h，Qのディメンジョン
1010：データをフロッピーに入力させるためのオープン文
1030：Qの増加式
1040：初期値
1050：hの初期値
1060：収束値
1080：Q，hの途中経過の出力
1090
～：P，G，P'，G'，$f_{(h)}$，$f'_{(h)}$，hの計算
1060
1170：$\lvert f_{(h)}\rvert$の判定
1180：hの修正式
1200：フロッピーにデータを書かせる文
1210：計算結果の出力
1220：$I=I+1$として次の流量へ
1230：フロッピーを閉じる



図 2 − 4 − 2　*h* − *Q* 曲線図